

# Microprocessors & Interfacing

## AVR Programming (II)

Lecturer : Dr. Annie Guo

## Lecture Overview

- Assembly program structure
  - Assembler directives
  - Assembler expressions
  - Macros
- Memory access
- Assembly process
  - First pass
  - Second pass

## Assembly Program Structure

- An assembly program basically consists of
  - Assembler directives
    - E.g. `.def temp = r15`
  - Executable instructions
    - E.g. `add r1, r2`
- An input line in an assembly program takes one of the following forms :
  - *[label:] directive [operands] [Comment]*
  - *[label:] instruction [operands] [Comment]*
  - *Comment*
  - *Empty line*

## Assembly Program Structure (cont.)

- *The label for an instruction is associated with the memory location address of that instruction.*
- *All instructions are not case sensitive*
  - *“add” is same as “ADD”*
  - *“.DEF” is same as “.def”*

## Example

```
; The program performs  
; 2-byte addition: a+b;
```

← *Two comment lines*

```
.def a_high = r2;  
.def a_low = r1;  
.def b_high = r4;  
.def b_low = r3;  
.def sum_high = r6;  
.def sum_low = r5;
```

← *Empty line*

← *Six assembler directives*

```
mov sum_low, r1  
mov sum_high, r3  
add sum_low, r2  
adc sum_high, r3
```

← *Four executable instructions*

## Comments

- A comment has the following form:
  - `;[Text]`
  - Items within the brackets are optional
- The text between the comment-delimiter(;) and the end of line (EOL) is ignored by the assembler.

## Assembly Directives

- Assembly directives are instructions to the assembler. They are created for a number of purposes:
  - For symbol definitions
    - For readability and maintainability
    - All symbols used in a program will be replaced by the real values during assembling
    - E.g. `.def`, `.set`
  - For program and data organization
    - E.g. `.org`, `.cseg`, `.dseg`
  - For data/variable memory allocation
    - E.g. `.DB`
  - For others

### Summary of AVR Assembler directives

Directive	Description
BYTE	Reserve byte to a variable
CSEG	Code Segment
DB	Define constant byte(s)
DEF	Define a symbolic name on a register
DEVICE	Define which device to assemble for
DSEG	Data Segment
DW	Define constant word(s)
ENDMACRO	End macro
EQU	Set a symbol equal to an expression
ESEG	EEPROM Segment
EXIT	Exit from file
INCLUDE	Read source from another file
LIST	Turn listfile generation on
LISTMAC	Turn macro expansion on
MACRO	Begin macro
NOLIST	Turn listfile generation off
ORG	Set program origin
SET	Set a symbol to an expression

NOTE: All directives must be preceded by **a period**

## Directives for Symbol Definitions

- **.DEF**

- Define symbols on **registers**

```
.DEF symbol = register
```

- E.g.

*.def temp=r17*

- Symbol *temp* can be used for r17 elsewhere in the program after the definition

## Directives for Symbol Definitions (cont.)

- **.EQU**

- Define symbols on **values**

```
.EQU symbol = expression
```

- E.g.

*.EQU length=2*

- Un-redefinable. The symbol cannot be redefined for other value in the program

- Symbol *length* with value 2 can be used elsewhere in the program after the definition

## Directives for Symbol Definitions (cont.)

- **.SET**

- Define symbols on **values**

```
.SET symbol = expression
```

- redefinable. The symbol can represent other value later.

- E.g.

*.set input=5*

- Symbol *input* with value 5 can be used elsewhere in the program after this definition and before its redefinition.

## Program/Data Memory Organization

- AVR has three different memories

- Data memory
- Program memory
- EPROM memory

- The three memories are corresponding to three memory segments to the assembler:

- Data segment
- Program segment (or Code segment)
- EEPROM segment

## Program/Data Memory Organization Directives

- Memory segment directives specify which memory segment to use
  - .DSEG**
    - Data segment
  - .CSEG**
    - Code segment
  - .ESEG**
    - EPROM segment
- The **.ORG** directive specifies the start address to store the related program/data.

## Example

```
.DSEG      ; Start data segment
.ORG 0x100 ; from address 0x100,
           ; default start location is 0x0060

varlab: .BYTE 4      ; Reserve 4 bytes in SRAM
           ; from address 0x100

.CSEG      ; Start code segment
           ; default start location is 0x0000

const: .DW 10, 0x10, 0b10, -1
           ; Write 10, 16, 2, -1 in program
           ; memory, each value takes
           ; 2 bytes.

mov r1,r0   ; Do something
```

## Data/Variable Memory Allocation Directives

- Specify the memory locations/sizes for
  - Constants**
    - In program/EEPROM memory
  - Variables**
    - In data memory
- All directives must start with a label so that the related data/variable can be accessed later.

## Directives for Constants

- Store data in **program/EEPROM memory**

### – .DB

- Store **byte** constants in program/EEPROM memory

```
Label: .DB    expr1, expr2, ...
```

– **expr\*** is a byte constant value

### – .DW

- Store **word** constants in program/EEPROM memory
- little endian** scheme is used

```
Label: .DW    expr1, expr2, ...
```

– **expr\*** is a word constant value

## Directives for Variables

- Reserve bytes in **data memory**
  - **.BYTE**
    - Reserve a number of bytes for a variable

```
Label: .BYTE  expr
```

- *expr* is the number of bytes to be reserved.

## Directives for Others

- Include a file
  - **.INCLUDE** "m64def.inc"
- Stop processing the assembly file
  - **.EXIT**
- Begin and end macro definition
  - **.MACRO**
  - **.ENDMACRO**
  - Will be discussed in detail later

## Implement Data/Variables

- With those directives, you can implement/translate data/variables into machine level descriptions
- An example of translation by WINAVR is given in the next slide.

## Sample C Program

```
// global variables:
const char g_course[ ] = "COMP";
char* g_inputCourse = "COMP";
char g_a;
static char g_b;

int main(void){
    // local variables:
    const char course[ ] = "COMP9032";
    char* inputCourse = "COMP9031";
    char a;
    static char b;
    char i;
    char isCOMP9032 = 1;

    for(i=0; i<9; i++){
        if (inputCourse[i] != course[i]){
            isCOMP9032 = 0;
            i = 9;
        }
    }
    return 0;
}
```


## Memory mapping after build and run

Name	Value	Type	Location
<b>g_course</b>	[...]	const char[5]	0x0100 [SRAM]
[0x00]	67 'C'	const char	0x0100 [SRAM]
[0x01]	79 'O'	const char	0x0101 [SRAM]
[0x02]	77 'M'	const char	0x0102 [SRAM]
[0x03]	80 'P'	const char	0x0103 [SRAM]
[0x04]	0 ''	const char	0x0104 [SRAM]
<b>g_inputCourse</b>	0x0105	char*	0x010A [SRAM]
->	67 'C'	char	0x0105 [SRAM]
<b>g_a</b>	0 ''	char	0x0120 [SRAM]
<b>g_b</b>	0 ''	char	0x011F [SRAM]
<b>course</b>	[...]	const char[9]	0x1100 [SRAM]
[0x00]	-1 'y'	const char	0x1100 [SRAM]
[0x01]	-1 'y'	const char	0x1101 [SRAM]
[0x02]	-1 'y'	const char	0x1102 [SRAM]
[0x03]	-1 'y'	const char	0x1103 [SRAM]
[0x04]	-1 'y'	const char	0x1104 [SRAM]
[0x05]	-1 'y'	const char	0x1105 [SRAM]
[0x06]	-1 'y'	const char	0x1106 [SRAM]
[0x07]	-1 'y'	const char	0x1107 [SRAM]
[0x08]	-1 'y'	const char	0x1108 [SRAM]
<b>inputCourse</b>	0xFFFF	char*	0x1109 [SRAM]
->	-1 'y'	char	0xFFFF [SRAM]
<b>a</b>	-1 'y'	char	0x110B [SRAM]
<b>b</b>	0 ''	char	0x011E [SRAM]
<b>i</b>	-1 'y'	char	0x110C [SRAM]

## Memory mapping after execution

Name	Value	Type	Location
<b>g_course</b>	[...]	const char[5]	0x0100 [SRAM]
[0x00]	67 'C'	const char	0x0100 [SRAM]
[0x01]	79 'O'	const char	0x0101 [SRAM]
[0x02]	77 'M'	const char	0x0102 [SRAM]
[0x03]	80 'P'	const char	0x0103 [SRAM]
[0x04]	0 ''	const char	0x0104 [SRAM]
<b>g_inputCourse</b>	0x0105	char*	0x010A [SRAM]
->	67 'C'	char	0x0105 [SRAM]
<b>g_a</b>	0 ''	char	0x0120 [SRAM]
<b>g_b</b>	0 ''	char	0x011F [SRAM]
<b>course</b>	[...]	const char[9]	0x10F2 [SRAM]
[0x00]	67 'C'	const char	0x10F2 [SRAM]
[0x01]	79 'O'	const char	0x10F3 [SRAM]
[0x02]	77 'M'	const char	0x10F4 [SRAM]
[0x03]	80 'P'	const char	0x10F5 [SRAM]
[0x04]	57 '9'	const char	0x10F6 [SRAM]
[0x05]	48 '0'	const char	0x10F7 [SRAM]
[0x06]	51 '3'	const char	0x10F8 [SRAM]
[0x07]	50 '2'	const char	0x10F9 [SRAM]
[0x08]	0 ''	const char	0x10FA [SRAM]
<b>inputCourse</b>	0x0115	char*	0x10FB [SRAM]
->	67 'C'	char	0x0115 [SRAM]
<b>a</b>	-1 'y'	char	0x10FD [SRAM]
<b>b</b>	0 ''	char	0x011E [SRAM]
<b>i</b>	10 '	char	0x10FE [SRAM]

## Memory Mapping Diagram



## Remarks

- Data have scope and duration in the program
- Data have types and structures
- Those features determine where and how to store data in memory.
- Constants are usually stored in the non-volatile memory and variables are allocated in SRAM memory.
- In this lecture, we will only take a look at how to implement basic data type.
  - Advanced data/variable implementation will be covered later.

## Example 1

- Translate the following C variables. Assume each integer takes four bytes.

```
int a;
unsigned int b;
char c;
char* d;
```

## Example 1: Solution

- Translate the following variables. Assume each integer takes four bytes.

```
.dseg          ; in data memory

.org 0x100     ; start from address 0x100

a: .byte 4     ; 4 byte integer
b: .byte 4     ; 4 byte unsigned integer
c: .byte 1     ; 1 character
d: .byte 2     ; address pointing to the string
```

- All variables are allocated in SRAM
- Labels are given the same name as the variable for convenience.

## Example 2

- Translate the following C constants and variables.

C code:

```
int a;
const char b[]="COMP9032";
const int c=9032;
```

Assembly code:

```
.dseg
.org 0x100
a: .byte 4

.cseg
b: .DB 'C', 'O', 'M', 'P', '9', '0', '3', '2', 0
c: .DW 9032
```

- All variables are in SRAM and constants are in FLASH

## Example 2 (cont.)

- An insight into the memory mapping
  - In program memory, data are packed in words. If only a single byte left, that byte is stored in high byte and the low byte is filled with 0.

		Hex values
0x0000	'C' 'O'	43 4F
0x0001	'M' 'P'	4D 50
0x0002	'9' '0'	39 30
0x0003	'3' '2'	33 32
0x0004	0 0	0 0
0x0005	9032	48 23

## Example 3

- Translate data structures

```
struct
{
    int student_ID;
    char name[20];
    char WAM;
} STUDENT_RECORD;

typedef struct STUDENT_RECORD *student;

student s1;
student s2;
```

## Example 3 : Solution

- Translate data structures

```
.set    student_ID=0
.set    name = student_ID+4
.set    WAM = name + 20
.set    STUDENT_RECORD_SIZE = WAM + 1

.dseg
s1:     .BYTE  STUDENT_RECORD_SIZE
s2:     .BYTE  STUDENT_RECORD_SIZE
```

## Example 4

- Translate data structures
  - with initialization

```
struct
{
    int student_ID;
    char name[20];
    char WAM;
} STUDENT_RECORD;

typedef struct STUDENT_RECORD *student;

student s1 = {123456, "John Smith", 75};
student s2;
```

## Example 4: Solution

- Translate data structures

```
.set    student_ID=0
.set    name = student_ID+4
.set    WAM = name + 20
.set    STUDENT_RECORD_SIZE = WAM + 1

.cseg
s1_value:    .DW    HWRD(123456)
              .DW    LWRD(123456)
              .DB    "John Smith"
              .DB    75

.dseg
s1:         .BYTE  STUDENT_RECORD_SIZE
s2:         .BYTE  STUDENT_RECORD_SIZE

; copy the data from instruction memory to s1
...
```



## Remarks


- The constant values for initialization are stored in the program memory in order to keep the values when power is off.
- The variable will be populated with the initial values when the program is started.

## Assembler Expression

- In the assembly program, you can use expressions for values.
- During assembling, the assembler evaluates each expression and replaces the expression with the related value.

## Assembler Expression (cont.)

- The expression is of the form similar to normal math expressions
  - Consisting of operands, operators and functions.All expressions are internally 32 bits.
- Example



## Operands

- Operands can be
  - User defined labels
    - associated with memory addresses
  - User defined variables
    - defined by the SET directive
  - User defined constants
    - defined by the EQU directive
  - Integer constants
    - in several possible formats, including
      - Decimal (default): 10, 255
      - Hexadecimal (two notations): 0x0a, \$0a, 0xff, \$ff
      - Binary: 0b00001010, 0b11111111
      - Octal (leading zero): 010, 077
  - PC
    - Program counter value.

## Operators

Same  
meanings  
as in C

Symbol	Description
!	Logical Not
~	Bitwise Not
-	Unary Minus
*	Multiplication
/	Division
+	Addition
-	Subtraction
<<	Shift left
>>	Shift right
<	Less than
<=	Less than or equal
>	Greater than
>=	Greater than or equal
==	Equal
!=	Not equal
&	Bitwise And
^	Bitwise Xor
	Bitwise Or
&&	Logical And
	Logical Or

S2, 2008

COMP9032 Week3

37

## Functions

- **LOW(expression)**
  - Returns the low byte of an expression
- **HIGH(expression)**
  - Returns the second (low) byte of an expression
- **BYTE2(expression)**
  - The same function as HIGH
- **BYTE3(expression)**
  - Returns the third byte of an expression
- **BYTE4(expression)**
  - Returns the fourth byte of an expression
- **LWRD(expression)**
  - Returns low word (bits 0-15) of an expression
- **HWRD(expression):**
  - Returns bits 16-31 of an expression
- **PAGE(expression):**
  - Returns bits 16-21 of an expression
- **EXP2(expression):**
  - Returns 2 to the power of expression
- **LOG2(expression):**
  - Returns the integer part of log2(expression)

S2, 2008

38

## Example 1

;Example1:

```
ldi r17, 1<<5 ;load r17 with 1
                ;shifted left 5 times
```

S2, 2008

COMP9032 Week3

39

## Example 2

;Example 2: compare r21:r20 with 3167

```
cpi r20, low(3167)
ldi r16, high(3167)
cpc r21, r16
brlt case1
```

```
...
case1: inc10
```

S2, 2008

COMP9032 Week3

40

## Macros

- A sequence of instructions in an assembly program often need to be repeated several times
- Macros help programmers to write code efficiently and nicely
  - Type/define a section code once and reuse it
    - Neat representation
  - like an inline function in C
    - When assembled, the macro definition is expanded at the place it was used.

## Detectives for Macros

- **.MACRO**
  - Tells the assembler that this is the start of a Macro
  - Takes the macro name and parameters
    - Up to 10 parameters
      - Which are referenced by @0, ...@9 in the macro definition body
- **.ENDMACRO**
  - Defines the end of a Macro definition.

## Macros (cont.)

- Macro definition structure:

```
.MACRO macro_name
    ;macro body
.ENDMACRO
```

- Use of Macro

```
macro_name [para0, para1, ...,para9]
```

## Example 1

- Swapping memory data p, q twice

### With macro

```
.macro swap1
    lds r2, p    ; load data
    lds r3, q    ; from p, q
    sts q, r2    ; store data
    sts p, r3    ; to q, p
.endmacro

swap1
swap1
```

### Without macro

```
lds r2, p
lds r3, q
sts q, r2
sts p, r3

lds r2, p
lds r3, q
sts q, r2
sts p, r3
```

## Example 2

- Swapping any two memory data

```
.macro swap2
    lds r2, @0      ; load data from provided
    lds r3, @1      ; two locations
    sts @1, r2      ; interchange the data and
    sts @0, r3      ; store data back
.endmacro

swap2 a, b          ;a is @0, b is @1
swap2 c, d          ;c is @0, d is @1
```

## Example 3

- Register bit copy
  - copy a bit from one register to a bit of another register

```
.macro bitcopy
    bst @0, @1
    bld @2, @3
.endmacro

bitcopy r4, 2, r5, 3
bitcopy r5, 4, r7, 6

end: rjmp end
```

## Memory Access Operations

- Access to data memory
  - Using instructions
    - ld, lds, st, sts
- Access to program memory
  - Using instructions
    - lpm
    - spm
    - Not covered in this course
  - Most of time, we access program memory is to load data

## Load Program Memory

- Syntax: *lpm Rd, Z*
- Operands:  $Rd \in \{r0, r1, \dots, r31\}$
- Operation:  $Rd \leftarrow (Z)$
- Words: 1
- Cycles: 3

## Load From Program Memory

- The address label in the memory program is **word address**
  - Used by the PC register
- To access data, the **byte address** is used.
- Address register, Z, is used to point bytes in the program memory

## Example

```
.include "m64def.inc"           ; include definition for Z

ldi ZH, high(Table_1<<1)       ; Initialize Z-pointer
ldi ZL, low(Table_1<<1)

lpm r16, Z                      ; Load constant from Program
                                ; memory pointed to by Z (r31:r30)

Table_1:
    .dw 0x5876                  ; 0x76 is the value when ZLSB = 0
                                ; 0x58 is the value when ZLSB = 1
```

## Complete Example 1

- Copy data from Program memory to Data memory

## Complete Example 1 (cont.)

- C description

```
struct
{
    int student_ID;
    char name[20];
    char WAM;
} STUDENT_RECORD;

typedef struct STUDENT_RECORD *student;

student s1 = {123456, "John Smith", 75};
```

## Complete Example 1 (cont.)

- Assembly translation

```
.set    student_ID=0
.set    name = student_ID+4
.set    WAM = name + 20
.set    STUDENT_RECORD_SIZE = WAM + 1

.cseg
s1_value:    .DW    HWRD(123456)
              .DW    LWRD(123456)
              .DB    "John Smith"
              .DB    75

start:    ldi r31, high(s1_value<<1)    ;pointer to student record
          ldi r30, low(s1_value<<1)     ;value in the program memory

          ldi r29, high(s1)              ;pointer to student record holder
          ldi r28, low(s1)               ;in the data memory

          clr r16
```

## Complete Example 1 (cont.)

- Assembly translation (cont.)

```
load:
          cpi r16, STUDENT_RECORD_SIZE
          brge end
          lpm r10, z+
          st y+, r10
          inc r16
          rjmp load

end:
          rjmp end

.dseg
.ORG 0x100
s1:    .BYTE    STUDENT_RECORD_SIZE
```

## Complete Example 2

- Convert lower-case to upper-case for a string
  - The string is stored in the program memory
  - The resulting string after conversion is stored in data memory.
  - In ASCII, upper case letter + 32 = low case letter

## Complete Example 2 (cont.)

- Assembly program

```
.include "m64def.inc"
.equ size =5
.def counter =r17
.dseg
.org 0x100                                ; Set the starting address
                                           ; of data segment to 0x100

Cap_string: .byte 5
.cseg
Low_string: .db "hello"
          ldi zl, low(Low_string<<1)    ; Get the low byte of
                                           ; the address of "h"
          ldi zh, high(Low_string<<1)    ; Get the high byte of
                                           ; the address of "h"

          ldi yh, high(Cap_string)
          ldi yl, low(Cap_string)

          clr counter                    ; counter=0
```

## Complete Example 2 (cont.)

- Assembly program (cont.)

```
main:
    lpm r20, z+ ; Load a letter from flash memory
    subi r20, 32 ; Convert it to the capital letter
    st y+,r20 ; Store the capital letter in SRAM
    inc counter
    cpi counter, size
    brlt main
loop: nop
    rjmp loop
```

## Assembly

- Assembly programs need to be converted to machine code before execution
  - This translation/conversion from assembly program to machine code is called **assembly** and is done by the **assembler**
- There are two general steps in the assembly processes:
  - Pass one
  - Pass two

## Two Passes in Assembly

- Pass One
  - Lexical and syntax analysis: checking for syntax errors
  - Expand macro calls
  - Record all the symbols (labels etc) in a symbol table
- Pass Two
  - Use the symbol table to substitute the values for the symbols and evaluate functions.
  - Assemble each instruction
    - i.e. generate machine code

## Example

Assembly program

```
.equ bound=5

    clr r10
loop:
    cpi r16, bound
    brlo end
    inc r10
    rjmp loop
end:
    rjmp end
```

Symbol table

Symbol	Value
bound	5
loop	1
end	5

## Example (cont.)


### Code generation

<u>Address</u>	<u>Code</u>	<u>Assembly statement</u>
00000000:	24AA	clr r10
00000001:	3005	cpi r16,0x05
00000002:	F010	brlo PC+0x03
00000003:	94A3	inc r10
00000004:	CFFC	rjmp PC-0x0003
00000005:	CFFF	rjmp PC-0x0000

## Absolute Assembly

- A type of assembly process.
  - Can only be used for the source file that contains all the source code of the program
- Programmers use .org to tell the assembler the starting address of a segment (data segment or code segment)
- Whenever any change is made in the source program, all code must be assembled.
- A loader transfers an **executable file** (machine code) to the target system.

## Absolute Assembly - workflow




## Relocatable Assembly

- Another type of assembly process.
- Each source file can be assembled separately
- Each file is assembled into **an object file** where some addresses may not be resolved
- A linker program is needed to resolve all unresolved addresses and make all object files into a single executable file



## Relocatable Assembly - workflow



S2, 2008

65

## Homework

- Refer to the AVR Instruction Set manual, study the following instructions:
  - Arithmetic and logic instructions
    - clr
    - inc, dec
  - Data transfer instructions
    - movw
    - sts, lds
    - lpm
    - bst, bld
  - Program control
    - jmp
    - sbrs, sbrc

S2, 2008

COMP9032 Week3

66

## Homework

- Design a checking strategy that can find the endianness of AVR machine.
- Discuss the advantages of using Macros. Do Macros help programmer write an efficient code? Why?

S2, 2008

COMP9032 Week3

67

## Homework

- Write an assembly program to find the length of a string. The string is stored in the program memory and the length will be stored in the data memory.

S2, 2008

COMP9032 Week3

68

## Homework

5. Write an assembly program to find the student average WAM in a class. The record for each student is defined as

```
struct
{
    int student_ID;
    char name[20];
    char WAM;
} STUDENT_RECORD;
typedef struct STUDENT_RECORD *student;
```

Assume there are 5 students and all records are stored in the program memory. The average WAM will be stored in the data memory.

## Reading Material

- Cady “Microcontrollers and Microprocessors”, Chapter 5
- User’s guide to AVR assembler
  - This guide is a part of the on-line documentations accompanied with AVR Studio. Click help in AVR Studio.